NAG Fortran Library Routine Document

G01GCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01GCF returns the probability associated with the lower tail of the non-central χ^2 distribution via the routine name.

2 Specification

```
real FUNCTION G01GCF(X, DF, RLAMDA, TOL, MAXIT, IFAIL)
INTEGER              MAXIT, IFAIL
real                 X, DF, RLAMDA, TOL
```

3 Description

The lower tail probability of the non-central χ^2 distribution with ν degrees of freedom and non-centrality parameter λ, $P(\mathrm{X} \leq x : \nu; \lambda)$, is defined by

$$P(X \leq x : \nu; \lambda) = \sum_{j=0}^{\infty} e^{-\lambda/2} \frac{(\lambda/2)^j}{j!} P(X \leq x : \nu + 2j; 0), \tag{1}$$

where $P(\mathrm{X} \leq x : \nu + 2j; 0)$ is a central χ^2 distribution with $\nu + 2j$ degrees of freedom.

The value of j at which the Poisson weight, $e^{-\lambda/2}\frac{(\lambda/2)^j}{j!}$, is greatest is determined and the summation (1) is made forward and backward from that value of j.

The recursive relationship:

$$P(X \leq x : a + 2; 0) = P(X \leq x : a; 0) - \frac{(x^a/2)e^{-x/2}}{\Gamma(a+1)} \tag{2}$$

is used during the summation in (1).

4 References

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

5 Parameters

1: X – ***real*** *Input*

On entry: the deviate from the non-central χ^2 distribution with ν degrees of freedom and non-centrality parameter λ.

Constraint: $\mathrm{X} \geq 0.0$.

2: DF – ***real*** *Input*

On entry: the degrees of freedom, ν, of the non-central χ^2 distribution.

Constraint: $\mathrm{DF} \geq 0.0$.

3: RLAMDA – ***real*** *Input*

On entry: the non-centrality parameter, λ, of the non-central χ^2 distribution.

Constraint: RLAMDA ≥ 0.0 if DF > 0.0 or RLAMDA > 0.0 if DF $= 0.0$.

4: TOL – ***real*** *Input*

On entry: the required accuracy of the solution. If G01GCF is entered with TOL greater than or equal to 1.0 or less than $10\times$ ***machine precision*** (see X02AJF), then the value of $10\times$ ***machine precision*** is used instead.

5: MAXIT – INTEGER *Input*

On entry: the maximum number of iterations to be performed.

Suggested value: 100. See Section 8 for further discussion.

Constraint: MAXIT ≥ 1.

6: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on exit IFAIL $= 1$, 2, 4 or 5, then G01GCF returns 0.0.

IFAIL $= 1$

On entry, DF < 0.0,
or RLAMDA < 0.0,
or DF $= 0.0$ and RLAMDA $= 0.0$,
or X < 0.0,
or MAXIT < 1.

IFAIL $= 2$

The initial value of the Poisson weight used in the summation (1) was too small to be calculated. The value of $P(\mathrm{X} \leq x : \nu; \lambda)$ is likely to be zero.

IFAIL $= 3$

The solution has failed to converge in MAXIT iterations.

IFAIL $= 4$

The value of a term required in (2) is too large to be evaluated accurately. The most likely cause of this error is both X and RLAMDA being very large.

IFAIL = 5

The calculations for the central χ^2 probability has failed to converge. This is an unlikely error exit. A larger value of TOL should be used.

7 Accuracy

The summations described in Section 3 are made until an upper bound on the truncation error relative to the current summation value is less than TOL.

8 Further Comments

The number of terms in (1) required for a given accuracy will depend on the following factors:

(i) The rate at which the Poisson weights tend to zero. This will be slower for larger values of λ.

(ii) The rate at which the central χ^2 probabilities tend to zero. This will be slower for larger values of ν and x.

9 Example

Values from various non-central χ^2 distributions are read, the lower-tail probabilities calculated, and all these values printed out, until the end of data is reached.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01GCF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             DF, PROB, RLAMDA, TOL, X
      INTEGER          IFAIL, MAXIT
*     .. External Functions ..
      real             G01GCF
      EXTERNAL         G01GCF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01GCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
      WRITE (NOUT,*) '     X       DF   RLAMDA   PROB'
      WRITE (NOUT,*)
      TOL = 0.5e-5
      MAXIT = 50
   20 READ (NIN,*,END=40) X, DF, RLAMDA
      IFAIL = -1
*
      PROB = G01GCF(X,DF,RLAMDA,TOL,MAXIT,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) X, DF, RLAMDA, PROB
      ELSE
         WRITE (NOUT,99999) X, DF, RLAMDA, PROB, ' NOTE: IFAIL = ',
     +     IFAIL
      END IF
      GO TO 20
   40 STOP
*
99999 FORMAT (1X,3F8.3,F8.4,A,I1)
      END
```

9.2 Program Data

```
G01GCF Example Program Data
  8.26   20.0   3.5            :X DF RLAMDA
  6.2     7.5   2.0            :X DF RLAMDA
 55.76   45.0   1.0            :X DF RLAMDA
```

9.3 Program Results

```
 G01GCF Example Program Results

      X       DF   RLAMDA   PROB

    8.260  20.000   3.500  0.0032
    6.200   7.500   2.000  0.2699
   55.760  45.000   1.000  0.8443
```